


Workaround the 16k Character Limit for EC2 User Data: Terraform Solution



Kevin De Notariis [Follow](#)

May 21 · 4 min read



In this brief article, I would like to explain a workaround on the 16k characters limit for the user data in an EC2 instance.

It is pretty straightforward and the bottom line is: upload your configuration files to an S3 bucket and grab them in the `user data` of the instance and run them.

Setup

I suppose you already have an AWS infrastructure governed by your Terraform code with an EC2 instance where you want to stick a long `user data` configuration.

The situation might be something like the following:

```
instance.tf ×
instance.tf > resource "aws_instance" "default" > tags > Name
1 resource "aws_instance" "default" {
2   ami = var.ami-id
3   iam_instance_profile = var.iam-instance-profile
4   instance_type = var.instance-type
5   key_name = var.key-pair
6   network_interface {
7     device_index = var.device-index
8     network_interface_id = var.network-interface-id
9   }
10
11   user_data = <<EOF
12   ## MY REALLY LONG USER DATA
13   .
14   .
15   .
16 EOF
17   tags = {
18     Name = var.name
19   }
20 }

variables.tf ×
variables.tf > variable "network-interface-id" > type
1
2 variable "ami-id" {
3   type = string
4 }
5
6 variable "iam-instance-profile" {
7   default = ""
8   type = string
9 }
10
11 variable "instance-type" {
12   type = string
13   default = "t2.micro"
14 }
15
16 variable "name" {
17   type = string
18 }
19
20 variable "key-pair" {
21   type = string
22 }
23
24 variable "device-index" {
25   type = number
26 }
27
28 variable "network-interface-id" {
29   type = string
30 }
```

Example of a setup

Namely we have a module defining an EC2 instance in a file called `instance.tf`, where its parameters are variables defined in a `variables.tf` file (kind of usual setup for terraform).

Supposing that the EC2 instance already has access to S3 buckets (otherwise you need to modify the `aws_iam_role` by adding the suitable policy) we can create an `s3.tf` file where we define our bucket that will store the config file(s) and we place the actual file:

```
s3.tf X
s3.tf > ...
1 resource "aws_s3_bucket" "user-data" {
2   bucket = "super-unique-bucket-name-of-your-choice"
3   acl = "private"
4 }
5
6 resource "aws_s3_bucket_object" "user-data" {
7   bucket = aws_s3_bucket.user-data.id
8   key = "user_data.sh"
9   source = "config/user_data.sh"
10  etag = filemd5("config/user_data.sh")
11 }
12
13
14
15
16
```

With these instructions, we create a bucket called `super-unique-bucket-name-of-your-choice` (which you should change to make it somewhat meaningful, but keeping in mind that it should be unique across all AWS) and we place there a `user_data.sh` file stored in a folder `config`.

If we had more than one bash script to run at creation time in the user data, we could upload all of them using the following:


```
# To upload all the config files in the folder jenkins_config


resource "aws_s3_bucket_object" "user-data" {
  bucket = aws_s3_bucket.user-data.id
  for_each = fileset("config/", "*")
  key = each.value
  source = "config/${each.value}"
  etag = filemd5("config/${each.value}")
}
```

To upload multiple script to run in the user data

Single User Data Script

For the simplest case of one `user_data.sh` script, the folder structure should then be something like:





folder structure

We need then to modify the `instance/instance.tf` `user_data` as follows:

```
resource "aws_instance" "default" {
  ami = var.ami-id
  iam_instance_profile = var.iam-instance-profile
  instance_type = var.instance-type
  key_name = var.key-pair
  network_interface {
    device_index = var.device-index
    network_interface_id = var.network-interface-id
  }

  user_data = <<EOF
#-----#
#-----> COPY THE CONFIG FILES FROM S3 <-----#
#-----#

sudo aws s3 cp s3://${var.bucket-config-name}/ ./ --recursive
sudo chmod +x user_data.sh

#-----#
#-----> RUN THE CONFIG FILES <-----#
#-----#

./user_data.sh

EOF
  tags = {
    Name = var.name
  }
}
```

New user data of the instance

Where we added the following directives:

```
sudo aws s3 cp s3://${var.bucket-config-name}/ ./ --recursive

sudo chmod +x user_data.sh

./user_data.sh
```

The first one will copy (`cp`) from the bucket specified in the variable `var.bucket-config-name` all its elements in the 'current directory' (`./`).

Then, we mark the file that will be downloaded (`user_data.sh`) as an executable.

Finally, we execute that script.

Clearly, we need to define a new variable in `instance/variables.tf` :

```
variable "ami-id" {
  type = string
}

variable "iam-instance-profile" {
  default = ""
  type = string
}

variable "instance-type" {
  type = string
  default = "t2.micro"
}

variable "name" {
  type = string
}

variable "key-pair" {
  type = string
}

variable "device-index" {
  type = number
}

variable "network-interface-id" {
  type = string
}


variable "bucket-config-name" {
  type = string
}
```

Which, in the `instance.tf` (in the root directory), should be populated with the name of the bucket, namely with the following directive:

```
bucket-config-name = aws_s3_bucket.user_data.id
```

Multiple User Data Scripts

In the case of multiple scripts, the folder structure should instead look like:



folder structure for more user data scripts

In this case, the new `user_data` of the instance, should then be modified as follows:

```
resource "aws_instance" "default" {
  ami = var.ami-id
  iam_instance_profile = var.iam-instance-profile
  instance_type = var.instance-type
  key_name = var.key-pair
  network_interface {
    device_index = var.device-index
    network_interface_id = var.network-interface-id
  }

  user_data = <<EOF
#-----#
#-----> COPY THE CONFIG FILES FROM S3 <-----#
#-----#

sudo aws s3 cp s3://${var.bucket-config-name}/ ./ --recursive
sudo chmod +x *.sh

#-----#
#-----> RUN THE CONFIG FILES <-----#
#-----#

./user_data01.sh
./user_data02.sh
./user_data03.sh

EOF
  tags = {
    Name = var.name
  }
}
```

new user data of the instance

Where we have added:

```
sudo aws s3 cp s3://${var.bucket-config-name}/ ./ --recursive

sudo chmod +x *.sh

./user_data01.sh

./user_data02.sh

./user_data03.sh
```

As before, we download the files in the current directory and then we change all of them (*.sh) to be executable.

After that, we simply run them.

Obviously, even in this case, we need to add the new variable `bucket-config-name` in the `variables.tf` in order to tell the instance where to download the script files.

Simple enough? Drop some feedback in the comments to let me know your thoughts and to help me improve the quality of these articles. Thanks!

More content at plainenglish.io

Terraform

AWS

Ec2

S3

Programming



[About](#) [Write](#) [Help](#) [Legal](#)

Get the Medium app

